


Viewing X-Sectional Results

Introduction

This tutorial was created using ANSYS 7.0 The purpose of this tutorial is to outline the steps required to view cross sectional results (Deformation, Stress, etc.) of the following example.



Preprocessing: Defining the Problem

1. **Give example a Title**
Utility Menu > File > Change Title ...
`/title, Cross-Sectional Results of a Simple Cantilever Beam`
2. **Open preprocessor menu**
ANSYS Main Menu > Preprocessor
`/PREP7`
3. **Create Block**
Preprocessor > Modeling > Create > Volumes > Block > By 2 Corners & Z
`BLC4, 0, 0, Width, Height, Length`
Where: Width: 40mm
 Height: 60mm
 Length: 400mm
4. **Define the Type of Element**
Preprocessor > Element Type > Add/Edit/Delete...
For this problem we will use the SOLID45 (3D Structural Solid) element. This element has 8 nodes each with 3 degrees of freedom (translation along the X, Y and Z directions).
5. **Define Element Material Properties**
Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic
In the window that appears, enter the following geometric properties for steel:
 - i. Young's modulus EX: 200000
 - ii. Poisson's Ratio PRXY: 0.3
6. **Define Mesh Size**
Preprocessor > Meshing > Size Cntrls > ManualSize > Global > Size
`esize, 20`
For this example we will use an element size of 20mm.
7. **Mesh the volume**
Preprocessor > Meshing > Mesh > Volumes > Free > click 'Pick All'
`vmesh, all`

Solution: Assigning Loads and Solving




1. **Define Analysis Type**
Solution > Analysis Type > New Analysis > Static
`ANTYPE, 0`
2. **Apply Constraints**

- Solution > Define Loads > Apply > Structural > Displacement > On Areas
Fix the left hand side (should be labeled Area 1).
3. **Apply Loads**
Solution > Define Loads > Apply > Structural > Force/Moment > On Keypoints
Apply a load of 2500N downward on the back right hand keypoint (Keypoint #7).
 4. **Solve the System**
Solution > (-Solve-) Current LS
SOLVE


Postprocessing: Viewing the Results

Now since the purpose of this tutorial is to observe results within different cross-sections of the colume, we will first outline the steps required to view a slice.

- Offset the working plane for a cross section view (**WPOFFS**)
- Select the **TYPE** of display for the section (**/TYPE**). For this example we are trying to display a section, therefore, options 1, 5, or 8 are relevant and are summarized in the table below.

Type	Description	Visual Representation
SECT or (1)	Section display. Only the selected section is shown without any remaining faces or edges shown	
CAP or (5)	Capped hidden display. This is as though you have cut off a portion of the model and the remaining model can be seen	
ZQSL or (8)	QSLICE Z-buffered display. This is the same as SECT but the outline of the entire model is shown.	

- Align the cutting plane with the working plane (**/CPLANE**)
1. **Deflection**
Before we begin selecting cross sections, let's view deflection of the entire model.
 - Select: **General Postproc > Plot Results > Contour Plot > Nodal Solu**



From this one may wish to view several cross sections through the YZ plane.

To illustrate how to take a cross section, let's take one halfway through the beam in the YZ plane

- First, offset the working plane to the desired position, halfway through the beam


Select: **Utility Menu > WorkPlane > Offset WP by Increments**

In the window that appears, increase Global X to 30 (Width/2) and rotate Y by +90 degrees


- Select the type of plot and align the cutting plane with the working plane (Note that in GUI, these two steps are combined)

Select: **Utility Menu > PlotCtrls > Style > Hidden-Line Options**

Fill in the window that appears as shown below to select `/TYPE=ZQSL` and `/CPLANE=Working Plane`



As desired, you should now have the following:



This can be repeated for any slice, however, note that the command lines required to do the same are as follows:

```

WPOFFS,Width/2,0,0      ! Offset the working plane for cross-section view
WPROTA,0,0,90          ! Rotate the working plane
/CPLANE,1              ! Cutting plane defined to use the WP
/TYPE,1,8
PLNSOL,U,SUM,0,1

```

Also note that to realign the working plane with the active coordinate system, simply use:
WPCSYS,-1,0


2. Equivalent Stress

Again, let's view stresses within the entire model.

First we need to realign the working plane with the active coordinate system. Select: **Utility Menu > WorkPlane > Align WP with > Active Coord Sys** (NOTE: To check the position of the WP, select Utility Menu > WorkPlane > Show WP Status)

Next we need to change /TYPE to the default setting(no hidden or section operations). Select: **Utility Menu > PlotCtrls > Style > Hidden Line Options...** And change the 'Type of Plot' to 'Non-hidden'


o Select: **General Postproc > Plot Results > Contour Plot > Nodal Solu > Stress > von Mises**



Let's say that we want to take a closer look at the base of the beam through the XY plane. Because it is much easier, we are going to use command line:

```
WPOFFS,0,0,1/16*Length    ! Offset the working plane
/CPLANE,1                 ! Cutting plane defined to use the WP
/TYPE,1,5                 ! Use the capped hidden display
PLNSOL,S,EQV,0,1
```

Note that we did not need to rotate the WP because we want to look at the XY plane which is the default). Also note that we are using the capped hidden display this time. You should now see the following:



3. Animation

Now, for something a little more impressive, let's show an animation of the Von Mises stress through the beam. Unfortunately, the ANSYS commands are not as user friendly as they could be... but please bear with me.

- Select: **Utility Menu > PlotCtrls > Animate > Q-Slice Contours**
- In the window that appears, just change the Item to be contoured to 'Stress' 'von Mises'
- You will then be asked to select 3 nodes; the origin, the sweep direction, and the Y axis. In the graphics window, select the node at the origin of the coordinate system as the origin of the sweep (the sweep will start there). Next, the sweep direction is in the Z direction, so select any node in the z direction (parallel to the first node). Finally, select the node in the back, bottom left hand side corner as the Y axis.

You should now see an animated version of the contour slices through the beam. For more information on how to modify the animation, type **help ancut** into the command line.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```
FINISH
/CLEAR

/Title, Cross-Sectional Results of a Simple Cantilever Beam
/PREP7

! All dims in mm
Width = 60
Height = 40
Length = 400

BLC4,0,0,Width,Height,Length ! Creates a rectangle

/ANGLE, 1 ,60.000000,YS,1 ! Rotates the display
/REPLOT,FAST ! Fast redisplay

ET,1,SOLID45 ! Element type

MP,EX,1,200000 ! Young's Modulus
MP,PRXY,1,0.3 ! Poisson's ratio

esize,20 ! Element size
vmesh,all ! Mesh the volume

FINISH
/SOLU ! Enter solution mode

ANTYPE,0 ! Static analysis
ASEL,S,LOC,Z,0 ! Area select at z=0
DA,All,ALL,0 ! Constrain the area
ASEL,ALL ! Reselect all areas

KSEL,S,LOC,Z,Length ! Select certain keypoint
KSEL,R,LOC,Y,Height
KSEL,R,LOC,X,Width
FK,All,FY,-2500 ! Force on keypoint
KSEL,ALL ! Reselect all keypoints

SOLVE ! Solve
FINISH
```

```
/POST1 ! Enter post processor

PLNSOL,U,SUM,0,1 ! Plot deflection
WPOFFS,Width/2,0,0 ! Offset the working plane for cross-section view
WPROTA,0,0,90 ! Rotate working plane
/CPLANE,1 ! Cutting plane defined to use the WP
/TYPE,1,8 ! QSLICE display

WPCSYS,-1,0 ! Deflines working plane location

WPOFFS,0,0,1/16*Length ! Offset the working plane
/CPLANE,1 ! Cutting plane defined to use the WP
/TYPE,1,5 ! Use the capped hidden display
PLNSOL,S,EQV,0,1 ! Plot equivalent stress

!Animation
ANCUT,43,0.1,5,0.05,0,0.1,7,14,2 ! Animate the slices
```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>